
      !
      ! Test program for TAMS 1701A switch controller
      ! (c) Tech Soft GmbH, Sven Henze
      ! www.TechSoft.de
      !
      ! Created:  24-Nov-2006
      ! Update:   24-Nov-2006
      !
      ! This program creates random frequencies with the TAMS 1701A frequency synthesizer
      !
      DIM Cmd$[80],A$[1024],Devname$[255]
      INTEGER Verbose
      LONG Cyc
      !
      Verbose=0
      !
      CONTROL KBD,203;1
      CONTROL KBD,204;1
      !
      RANDOMIZE
      !
      CONTROL CRT,100;1
      !
      G$=CHR$(19)                         ! character code for inverse printing
      X$=CHR$(16)                         ! character code for normal printing
      Hl$=CHR$(26)                        ! character code for "highlighted" printing
      Hx$=CHR$(24)                        ! character code for "normal" printing
      Ul$=CHR$(22)                        ! character code for underlined printing
      Crlf$=CHR$(10)&CHR$(13)
      !
      IF FNHtbvers>=9.4 THEN              ! only possible with HTBasic 9.4 and higher
          Devname$="UsbDevice1"
          ON ERROR GOTO Drv_load_err
          !
          IF SYSTEM$("VERSION:USBS")="0" THEN LOAD BIN "USBS;DEV "&Devname$&" TIME 10"
          CLEAR SCREEN
          !
          PRINT G$&" TAMS 1701A test program "&X$
          PRINT
          !
          Dummy=FNSend_command(7,"*IDN?") ! returns identifier
          PRINT " "&Ul$&"Device ID"&X$&":     ";FNRead_device$(7)
          !
          Dummy=FNSend_command(7,"*RST")  ! reset device
          !
          Dummy=FNSend_command(7,"*TST?")
          PRINT " "&Ul$&"Test result"&X$&":   ";FNRead_device$(7)
          !
          STATUS 7,1;Reg1                 ! does device implement the USBTMC protocol?
          STATUS 7,2;Reg2                 ! is TRIGGER supported?
          STATUS 7,3;Reg3                 ! is REMOTE supported?
          STATUS 7,4;Reg4                 ! are SCPI commands suoported?
          PRINT " "&Ul$&"Register 1..4"&X$&": ";Reg1;"  ";Reg2;"  ";
          PRINT Reg3;"  ";Reg4;"  (USBTMC, Trigger, Remote, SCPI)"
          PRINT
          !
          REAL Waittime=5                 ! wait this time before sending new frequency
          Maxfreq=50000                   ! maximum generated frequency (in Hz)
          !
          Cyc=0
          R$=VAL$(PROUND(Waittime,-3))&" seconds   "
          IF R$[1;1]="." THEN R$="0"&R$
          PRINT " "&Ul$&"Frequency switch range"&X$&":  0.05 .. ";VAL$(Maxfreq)&" Hz"
          PRINT " "&Ul$&"Frequency switch delay"&X$&":  ";R$
          PRINT
          Sl=10
          !
          User_break=0
          ON KBD ALL GOSUB Kbd_break
          DISP Hl$&"Press any key to stop the program"&Hx$

          !
          Faktor=10
          WHILE NOT User_break
              !
              IF User_break THEN I=Numloops
              Cyc=Cyc+1
              !
              Freq=Maxfreq*RND+.05
              !
              IF NOT User_break THEN D=FNSend_command(7,"FREQ "&VAL$(Freq)&"HZ;*OPC?")
              ! read returned status info
              IF NOT User_break THEN A$=FNRead_device$(7)
              !
              ! initiate read command
              IF NOT User_break THEN D=FNSend_command(7,"FREQ ?")
              ! read back the settled frequency
              IF NOT User_break THEN A$=FNRead_device$(7)
              !
              BEEP 2000,.01
              PRINT TABXY(2,10);Ul$;"Loop number"&X$&":          "&VAL$(Cyc)&"      "
              B$=VAL$(PROUND(Freq,-2))                    ! some rounding
              PRINT TABXY(2,11);Ul$;"Frequency wanted"&X$&":     "&B$&" Hz     "
              PRINT TABXY(2,12);Ul$&"Frequency generated"&X$&": "&A$&" Hz        "
              !
              IF NOT User_break THEN GOSUB My_wait
              !
          END WHILE
      ELSE
          BEEP
          PRINT
          ALPHA PEN 2
          PRINT " ** ERROR: Wrong HTBasic version **"
          ALPHA PEN 1
          PRINT " Sorry, you are using HTBasic version "&VAL$(FNHtbvers);
          PRINT " which is not compatible with this demo program."
          PRINT " This example only works with HTBasic 9.4 and above."
          PRINT
          PRINT " Program stopped."
          STOP
      END IF
      !
      DISP "Program finished."
      !
      STOP
      !
 My_wait: ! special wait routine
      Granul=20
      IF Waittime<1 THEN
          WAIT Waittime
      ELSE
          FOR I_=1 TO Waittime STEP Waittime/Granul
              IF User_break THEN I_=Waittime   ! leave wait routine if user break occured
              WAIT Waittime/Granul
          NEXT I_
      END IF
      RETURN
      !
 Kbd_break:!
      BEEP
      User_break=1
      DISP "User break. ";
      RETURN
      !
 Drv_load_err:  ! error during driver loading
      SELECT ERRN
      CASE 167           ! Interface Status Error
          BEEP
          ALPHA PEN 2
          PRINT "** ERROR 167 while loading USBS driver"
          ALPHA PEN 1



Page 3/3

          PRINT "    - Is the device connected and blinking?"
          PRINT "    - Is the devicename configured correctly ('"&Devname$&"')?"
      CASE 731
          BEEP
          ALPHA PEN 2
          PRINT "** ERROR 731 while talking to the SICL driver"
          ALPHA PEN 1
          PRINT "    - Is the device still connected and blinking?"
      END SELECT
      PRINT
      PRINT "Program stopped."
      STOP
      !
      END
      !
      DEF FNHtbvers
      !
          REAL Htbvers
          DIM Htbvers$[80]
          !
          Htbvers$=SYSTEM$("VERSION:HTB")
          !
          ! version number is located 16 characters after the occurance of "Release"
          Htbvers=VAL(Htbvers$[(POS(Htbvers$,"Release"))+8;3])
          !
          RETURN Htbvers
          !
      FNEND
      !
      DEF FNSend_command(INTEGER Isc,Cmd$,OPTIONAL INTEGER Verbose)
      !
          Verb_=0
          IF NPAR>=3 THEN
              Verb_=Verbose
          END IF
          !
          ON TIMEOUT Isc,5 GOTO Sc_timeout
          !
          ! make sure END is specified, otherwise the system may hang
          OUTPUT Isc;Cmd$ END
          IF Verb_ THEN PRINT Cmd$&"  ",
          !
          RETURN 1
          !
 Sc_timeout:         !
          BEEP
          PRINT "** Timeout ERROR in FNSend_command **"
          RETURN 0
      FNEND
      !
      DEF FNRead_device$(INTEGER Isc)
          DIM A$[1024]
          !
          ON TIMEOUT Isc,5 GOTO Rd_timeout
          !
          ENTER Isc;A$
          !
          RETURN A$
          !
 Rd_timeout:     !
          BEEP
          PRINT "** Timeout ERROR in FNRead_device **"
          RETURN ""
      FNEND


